The Impact of Clouds on Hourly Solar Radiation on Prairie

By

Xuejun Dong, Ph.D.

Distributed by

InfoClearinghouse.com

The impacts of clouds on hourly solar radiation on prairie

Xuejun Dong
Central Grasslands Research Extension Center, Streeter, ND
February 25, 2005

This note records what I did in calculating the influence of clouds on hourly variations in solar radiation, based on past weather data. I planned to find out, on an average non-rainy day, during the 6 high radiation hours from 10 am to 4 pm, what is the probability that radiation changes at least once between two adjacent hours. First we assume that a more than 30 % change in radiation (increase or decrease) during these high light hours of a day would likely represent the influence of clouds.

Hourly data of solar radiation and precipitation were downloaded from the NDAWN web site. The data available for Streeter Station is for 15 years. The first step is to calculate the probability that a 30 % radiation change happened between two adjacent hours from 10 am to 4 pm for the past 15 years at Streeter Station. The data processing was done using MINITAB. For each years's data from May to September, assume the original data is stored exactly in the columns of a MINITAB worksheet as shown in Table 1. There are 4 steps to get this calculation done:

1. First we use macro *avrg1.txt* to calculate the accumulated amount of rainfall on each day.
2. Then, we use macro *avrg2.txt* to pick out days having an accumulated rainfall of more than 0.02 in. These are days that will be considered as "non-rainy days".
3. The hourly data on each day between 10 am and 4 pm that changed at more than 30% when compared with the next hour's radiation data is extracted from the original data set using macro *prob1.txt*.
4. The counts of number of all the hourly data on non-rainy days in each month having a more than 30 % change in radiation was recorded using macros *prob2.txt* and *prob3.txt*.

For each month (May through September), let the counts of number of a major between-hour radiation change on non-rainy days in year i be N_i, and let the number of days in that month be D_{total} and number of rainy days in year i be $D_{rain,i}$, the probability, p, that a 30 % radiation change happened between two adjacent hours from 10 am to 4 pm on non-rainy days for the past 15 years is calculated as,

$$p = \frac{\sum_{i=1}^{15} N_i}{\sum_{i=1}^{15}[(D_{total} - D_{rain,i}) \times 6]} \tag{1}$$

There are 2 possibilities for the radiation change between any of the 2 adjacent hours (for example, from 10 am to 11 am, etc.): either the change is greater than 30 %, or it is equal to or less than 30 %. Here we are interested in the greater than 30% change. Let the event that such kind of change happens at least once for the 6 intervals be denoted as $gt1$ and its probability $p(gt1)$. For simplicity, the outcomes for each of the

Table 1: Format of original MINITAB data set of hourly radiation and precipitation from May to September in one year. To use the MINITAB macros, data must be stored in the columns (abbreviated as "C") as shown here.

$C1$	$C2$	$C3$	$C4$	$C5$	$C6$	$C7$
Station	*Year*	*Month*	*Day*	*Hour*	*Radiation*	*Precipitation*
Streeter	1990	5	1	12	50	0.0
.	.	.	.	.	.	.
.	.	.	.	.	.	.
.	.	.	.	.	.	.

6 intervals (from 10 am to 4 pm) can be assumed to be mutually independent. Then, we can calculate the probability as

$$p(gt1) = \sum_{i=1}^{6} \binom{6}{i} p^i (1-p)^{6-i} \tag{2}$$

This is the sum of the probability that such kind of change happens for exactly once, twice, three times, ..., and six times. In the computer, it is easy to do such additions, but if we prefer to using pencil and paper, it is faster to calculate the above probability by using the formula that set A equals one minus its compliment (A=1-$\bar{A}$),

$$p(gt1) = 1 - \binom{6}{0} p^0 (1-p)^{6-0} = 1 - (1-p)^6 \tag{3}$$

in which the term

$$p(0) = \binom{6}{0} p^0 (1-p)^{6-0} \tag{4}$$

is the probability that such kind of change does not happen at all (happens for zero times). The results are listed in Table 2. The binomial coefficient is calculated as

$$\binom{6}{0} = \frac{6!}{0!(6-0)!} = 1 \tag{5}$$

From the results of Table 2, we can predict that, if we go out on an average non-rainy June day at Streeter, we are 73% sure that the high light period of the day (from 10 am to 4 pm) would have a significant radiation change for at leat once. However, on August, an average non-rainy day would more likely to be perfectly clear without major influence from the patchy clouds when compared with June. Although the data analysis is very rough, some implications can be pointed out. Whether people like a perfectly clear sky depends on different things. A perfectly clear day is good for the crops under good water supply to maximize photosynthesis. However, if our crops have been under drought recently, we might want some clouds come to mitigate the drought stress. Or if we want to go out to watch birds, or go fishing, we may not want the non-rainy days we choose to be perfectly clear either, but we hope the days be comfortable with the decoration of patchy clouds.

Table 2: Probability of hourly radiation change during the high light hours from 10 am to 4 pm in past 15 years (1990-2004) at Streeter, ND. p is the probability that one such radiation change happens between adjacent 2 hours from 10 am to 4 pm for a particular month; $p(gt1)$ is the probability that such kind of change happens at least once for the 6-hour period from 10 am to 4 pm in each month.

Month	p	$p(gt1)$
May	0.178	0.69
June	0.194	0.73
July	0.169	0.67
August	0.166	0.66
September	0.177	0.69

Figure 1: *A summer day on the prairie of the Missouri Coteau. Picture by Rick Bohn.*

1 MINITAB macros

1.1 Marco *avrg1.txt*

```
gmacro
avrg1

Note How many rows of data do you have for this sheet?
Set C90;
  FILE "TERMINAL";
  NOBS 1.
COPY C90 k90   # k90 is the number of rows in this data sheet
erase c90

let k3=1
do k1=1:k90
  let k2=k1+1
  if c4(k2)=c4(k1)
    let k3=k3+1
    next
  elseif c4(k2)='*' # add "*" in the immediate following row of c6
    call mult_obs   # but delete it after run the macro
    erase c90       # This is just to make the program run OK
    erase k1-k4 k8 k10 k11 k90   # erase the intermediate variables
    exit
  else
    call mult_obs     # case for multiple measurements
    erase c90         # important: to be prepared to future use
    let k3=1
    next
  endif
enddo erase k1-k4 k8 k10 k11 k90  # erase the intermediate
endmacro # main macro ends


# subroutine followed
gmacro
 mult_obs
#sum c7 and store to c52
# letc50(1)=*, c51(1) = *, c52(1)=*
  let k4=k1-k3+1
  let k10=count(c50)+1
  let k11=1
  do k8=k4:k1
    let c90(k11)=c7(k8)
    let k11=k11+1
  enddo
```

```
  let c50(k10)=c3(k1)
  let c51(k10)=c4(k1)
  let c52(k10)=sum(c90)
  endmacro
```

1.2 Macro *avrg2.txt*

```
gmacro
avrg2

Note How many rows of data do you have for this sheet?
Set C90;
  FILE "TERMINAL";
  NOBS 1.
COPY C90 k90   # k90 is the number of rows in this data sheet
erase c90

do k1=1:k90
  if c52(k1)>0.02
  call rainday
  endif
enddo
endmacro

gmacro
 rainday
 let k2=count(c53)+1
 let c53(k2)=c50(k1)
 let c54(k2)=c51(k1)
 endmacro
```

1.3 Macro *prob1.txt*

```
gmacro
prob1

# To run this macro, first set the first row from c9-c14 to "*";
 #then set the cell just following c5 to "*";

 # also set the cell at the same row in c14 to "a". This is just to make
  # the macro run OK.

Note How many rows of data do you have for this sheet?
Set C90;
  FILE "TERMINAL";
```

```
  NOBS 1.
COPY C90 k90   # k90 is the number of rows in this data sheet
erase c90

name c9 'Y',c10 'Mth', c11 'dy', c12 'hr', c13 'rad',

name c14 'pct-diff'

do k1=1:k90
  if c5(k1)>=1000 & c5(k1)<=1500
    let k2=(c6(k1+1)-c6(k1))/c6(k1)
    let k66=abs(k2)
    if k66>=0.3
      call cloud
    endif
  endif
enddo
endmacro

gmacro
 cloud
 let k3=count(c9)+1
 let c9(k3)=c2(k1)

 let c10(k3)=c3(k1)
 let c11(k3)=c4(k1)
 let c12(k3)=c5(k1)
 let c13(k3)=c6(k1)
 let c14(k3)=k2

endmacro
```

1.4 Macro *prob2.txt*

```
gmacro prob2

# Important note: If the number of rows in c53 is not 47, then we
# have to change it accordingly.

Note How many rows of data do you have for this sheet?
Set C90;
  FILE "TERMINAL";
  NOBS 1.
COPY C90 k90   # k90 is the number of rows in this data sheet
erase c90
```

```
name c15 'RainOrNot'

do k1=1:k90
  let k2=c10(k1)
  let k3=c11(k1)
  do k4=1:47     # If the number of rows of c53 is not 47, change it
    if c53(k4)=k2 & c54(k4)=k3   # accordingly.
      let c15(k1)="Rainday"
    endif
  enddo
enddo

do k1=1:k90
  if c15(k1)<>"Rainday"
    let c15(k1)="NR"
  endif
enddo

endmacro
```

1.5 Macro *prob3.txt*

```
gmacro
prob3

# The results will be printed out on the monitor

Note How many rows of data do you have for this sheet?
Set C90;
  FILE "TERMINAL";
  NOBS 1.
COPY C90 k90   # k90 is the number of rows in this data sheet
erase c90

let k3=1
 do k1=1:k90
  let k2=k1+1
  if c10(k2)=c10(k1)
    let k3=k3+1
    next
  elseif c10(k2)='*'    # add "*" in the immediate following row of c6
    call mult_obs   # but delete it after run the macro
    erase c90       # This is just to make the program run OK
    erase k1-k8 k90  # erase the intermediate variables
    exit
  else
    call mult_obs     # case for multiple measurements
```

```
    erase c90         # important: to be prepared to future use
    let k3=1
    next
  endif
enddo erase k1-k8 k90 # erase the intermediate variables endmacro
# main macro ends


# subroutine followed
gmacro
mult_obs

let k4=k1-k3+1
 let k5=0
 do k7=k4:k1
 if c15(k7)="Rainday"
   let k5=k5+1
 endif
enddo

let k6=k3-k5
 Note Times of between-hour radiation change for

 Note non-rainy days for the month:
  name k8 'MonthofYear', k6 'Times_Rad_Change'
let k8=c10(k4)
print k8,k6
endmacro
```